
Invoking
The Natural program NOPUT1-P can be invoked to find out how API routines react to real data. It shows how to
code the calls of the supplied routines, and how to check the return codes.

This is the front end of NOPUT1-P:

 Entire Operations Application Program Interface           
                                                         
 Please Select:                                           
   Network / Job Status       ST                  NOPUST1N
   Above + wait for / error   SW                  NOPUST2N
   Conditions                 CO                  NOPUCN3N
   Resource Master            RM                  NOPURS1N
   Resource Usage             RU                  NOPURE1N
   Symbols                    SY                  NOPUSY4N
   Calendars/Schedules        CS                  NOPUCS1N
   Activation                 AC                  NOPUAC4N
   Job Import                 JI                  NOPUJI4N
   Version Info               VE                  NOPUVI2N
   Symbol Prompt Texts        SP                  NOPUSP1N
   Job Schedule               SC                  NOPUJS1N
   Log Monitor Activity       LM                  NOPMLA1N
   Active Job Statistics      AS                  NOPUAS1N
   Blank or . = End                                       
               ===> __

Source Code
The following is a list of the program NOPUT1-P contained in the Entire Operations library.

 
* NOPUT1-P:

*   Entire Operations: TEST FOR USER ACCESS ROUTINES

*

* FUNCTION:

*   THE Entire Operations USER ACCESS ROUTINES CAN BE INVOKED

*   FROM HERE, AND THEIR BEHAVIOUR CAN BE OBSERVED ON REAL DATA.

*   THE PROGRAM IS MENU-DRIVEN.

*

* MODIFICATIONS:

* 16.02.00 (192108) Function S (resubmit)                   XSISAA321

* 24.02.00 (190708.2) NOPMLA1N                                  SN321

* 01.03.00 (192559) NOPUAC4N Fct I (set run number start)       SN321

* 08.11.00 (202401) Change start time of Active job         XSISIB321Z2

* 14.03.01 207190   Status of next Job waiting or error     XSETIB411

* 08.06.01 (080601) Active job statistics (NOPUAS1N)           SJU411

* 15.06.01 (210679) NOPUJI4N                                   SJU411

* 17.08.01 E41102   NOPURE1N                                    SN411

* 23.11.01 206882.1 NOPUCN3N                                    SN411

* 08.11.01 E41116.3 Reserved name prefix =EOR=              XSETIB411

* ---------------------------------------------------------------------

DEFINE DATA LOCAL

* -- PARAMETERS FOR FUNCTIONS --

1 C-FUNC           (A1)
1 C-RC             (N3)
1 C-DDBID          (I2)
1 C-DFNR           (I2)
1 C-OBJECT-TYPE    (A1)
1 C-DBENV          (A10)
1 C-OWNER          (A10)
1 C-NETWORK        (A10) INIT <’ ’>
1 C-JOB            (A10) INIT <’ ’>           /* (E-0001)
1 C-RUN            (P13)
1 C-RUN-TO         (P13)                      /* 206882.1
1 C-RUN-4          (I04)
1 C-JOB-ID-10      (A10) INIT <’ ’>           /* 011023
1 C-OS-JOB-NAME    (A8)  INIT <’ ’>
1 C-EXECUTION-NODE (N3)  INIT <0>             /* (E-0005)
1 C-MOD-USER       (A8)
1 C-MOD-TIME       (T)
1 C-START-TIME     (T)
1 C-CONDITION      (A20)
1 C-COND-TIME-FROM (T)             /* (E-0003)
1 C-COND-TIME-TO   (T)             /* (E-0003)
1 C-SYMBOL-TABLE   (A10)
1 C-SYMBOL-NAME    (A20)
1 C-SYMBOL-FORMAT  (A1)
1 C-SYMBOL-VALUE   (A80)           /* V (111755)
1 REDEFINE C-SYMBOL-VALUE
  2 C-SYMBOL-VALUE-1   (A40)
  2 C-SYMBOL-VALUE-2   (A40)       /* A (111755)
1 C-OBJECT         (A10)           /* (117624)
1 C-DATE-A8        (A08)           /* (117624)
1 C-RESOURCE       (A20)           /* (109309)
1 C-RESOURCE-TYPE  (A01)           /* (109309)
1 C-INIT-QTY-P     (P07.2)         /* (109309)
1 C-QTY-P          (P07.2)         /* (109309)
1 C-QTY-N          (N07.2)
1 C-VERSION        (A08)           /* (920918)
1 C-VERSION-DATE   (A08)           /* (920918)

1 C-UPDATE-DATE    (A08)           /* (980728)
1 C-PROMPT-TEXT    (A70/1:5)       /* (146247)
1 C-EARLIEST-START (A14)           /* (183226)
1 REDEFINE C-EARLIEST-START
  2 #EARLIEST-START (T)
1 C-LATEST-START   (A14)           /* (183226)
1 C-DEADLINE       (A14)           /* (183226)
1 C-AVERAGE-TIME   (I04)           /* (183226)
1 C-STATUS-TIME    (A14)           /* 207190
1 C-BS2000-MONJV   (A250)          /* SJU 01-06-15 (210679) V
1 REDEFINE C-BS2000-MONJV
  2 C-BS2000-MONJV-54 (A54)
1 C-SYSOUT-FILE    (A250)
1 REDEFINE C-SYSOUT-FILE
  2 C-SYSOUT-FILE-54  (A54)
1 C-JI-EXEC-NODE   (I04)
1 C-SYSOUT-NODE    (I04)           /* SJU 01-06-15 (210679) A
1 C-DEALLOCATION   (A01)
1 C-CONFIRM        (A01)           /* 206882.1
1 C-JCL-CHECK      (A01)           /* 206882.1
1 C-JOB-EXECUTED   (L)             /* E41102
*
1 #START-TIME      (A14)           /* YYYYMMDDHHIISS
1 #DATE-FROM       (D)             /* (E-0003)
1 #DATE-TO         (D)             /* (E-0003)
1 #OP              (A2)
1 #RT              (A50)
1 #LFILE-SET       (L)       INIT *
* -- NAMES OF USER ROUTINES --
*
1 #PRG-SYM  (A8) CONST <’NOPUSY4N’>
1 #PRG-COND (A8) CONST <’NOPUCN3N’>          /* 206882.1
1 #PRG-LOG  (A8) CONST <’NOPULW4N’>
1 #PRG-SDA  (A8) CONST <’NOPUSD2N’>
1 #PRG-ACT  (A8) CONST <’NOPUAC4N’>
1 #PRG-JI   (A8) CONST <’NOPUJI4N’>
1 #PRG-CS   (A8) CONST <’NOPUCS1N’>
1 #PRG-RSC  (A8) CONST <’NOPURS1N’>
1 #PRG-RU   (A8) CONST <’NOPURE1N’>          /* E41102
1 #PRG-VERS (A8) CONST <’NOPUVI2N’>
1 #PRG-STAT (A8) CONST <’NOPUST1N’>
1 #PRG-PT   (A8) CONST <’NOPUSP1N’>
1 #PRG-SC   (A8) CONST <’NOPUJS1N’>          /* (183226)
1 #PRG-MLA  (A8) CONST <’NOPMLA1N’>          /* (190708.2)
1 #PRG-STE  (A8) CONST <’NOPUST2N’>          /* 207190
1 #PRG-AJS  (A8) CONST <’NOPUAS1N’>          /* (080601)

LOCAL USING NOPUAS1L                         /* (080601)

END-DEFINE

* -------------------------------------------------------------------

R-SEL. REPEAT
  INPUT ’Entire Operations Application Program Interface’(I)
    // ’Please Select:’
    /  3T ’Network / Job Status’
    30T ’ST’(I) 50T #PRG-STAT (AD=OD IP=OFF)
    /  3T ’Above + wait for / error’                   /* 207190
    30T ’SW’(I) 50T #PRG-STE  (AD=OD IP=OFF)           /* 207190
    /  3T ’Conditions’    30T ’CO’(I) 50T #PRG-COND (AD=OD IP=OFF)
    /  3T ’Resource Master’

    30T ’RM’(I) 50T #PRG-RSC  (AD=OD IP=OFF)
    /  3T ’Resource Usage’
    30T ’RU’(I) 50T #PRG-RU   (AD=OD IP=OFF)
    /  3T ’Symbols’       30T ’SY’(I) 50T #PRG-SYM  (AD=OD IP=OFF)
    /  3T ’Calendars/Schedules’
    30T ’CS’(I) 50T #PRG-CS   (AD=OD IP=OFF)
    /  3T ’Activation’    30T ’AC’(I) 50T #PRG-ACT  (AD=OD IP=OFF)
    /  3T ’Job Import’    30T ’JI’(I) 50T #PRG-JI   (AD=OD IP=OFF)
    /  3T ’Version Info’  30T ’VE’(I) 50T #PRG-VERS (AD=OD IP=OFF)
    /  3T ’Symbol Prompt Texts’ 30T ’SP’(I) 50T #PRG-PT   (AD=OD IP=OFF)
    /  3T ’Job Schedule’  30T ’SC’(I) 50T #PRG-SC   (AD=OD IP=OFF)
    /  3T ’Log Monitor Activity’
    30T ’LM’(I) 50T #PRG-MLA  (AD=OD IP=OFF)

* sju 01-06-08 next line added (080601)

    /  3T ’Active Job Statistics’ 30T ’AS’(I) 50T #PRG-AJS(AD=OD IP=OFF)
    /  3T ’Blank or . = End’
    /  15T ’===>’ #OP (AD=AIT’_’)
*
  IF #OP = ’ ’ OR = ’.’  ESCAPE ROUTINE  END-IF
*
  DECIDE ON FIRST VALUE OF #OP
    VALUE ’ ’  ESCAPE ROUTINE
*
    VALUE ’AC’       /* -- ACTIVATION --
      RESET C-RC #RT
      R-ACT. REPEAT
        INPUT ’Activation’(I) #PRG-ACT  (AD=OD IP=OFF)
          // ’Function       ==> ’ C-FUNC (AD=MIT’_’)
          ’(A,1 activate, H act. with hold, R,2 release,’
          / 24T ’D deactivate, X existence test, S resubmit,’  /* 192108
          / 24T ’C change activation time, I run number start,’
          / 24T ’J change job start time, ’’ ’’ end)’      /* 202401
          /  ’Owner          ==> ’ C-OWNER         (AD=MIT’_’)
          /  ’Network        ==> ’ C-NETWORK       (AD=MIT’_’)
          /  ’Job            ==> ’ C-JOB           (AD=MIT’_’)
          /  ’Run Number     ==> ’ C-RUN   (AD=MILT’_’ ZP=OFF)
          ’(Functions J,R)’                /* 202401
          /  ’Symbol Table   ==> ’ C-SYMBOL-TABLE  (AD=MIT’_’)
          /  ’Start Time     ==> ’ #START-TIME     (AD=MIT’_’)
          ’(YYYYMMDDHHIISS)’
          // ’Return Code    ==> ’ C-RC  (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’ ESCAPE BOTTOM (R-ACT.)  END-IF
*       IF #START-TIME NE ’ ’           /* V (111107)
        IF #START-TIME = MASK(YYYYMMDD00:2300:5900:59)   /* 188167
          MOVE EDITED #START-TIME TO C-START-TIME (EM=YYYYMMDDHHIISS)
        ELSE
          RESET C-START-TIME
        END-IF                          /* A (111107)
        CALLNAT #PRG-ACT
          C-FUNC C-RC C-OWNER C-NETWORK C-JOB C-SYMBOL-TABLE
          C-START-TIME C-RUN
        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE   0 MOVE ’ok’                         TO #RT
          VALUE   1 MOVE ’owner, net, job not found’  TO #RT
          VALUE   2 MOVE ’activation entry not found’ TO #RT
          VALUE   3 MOVE ’network, job not active’    TO #RT
          VALUE   4                                      /* 180719.2
            MOVE ’job can not be resubmitted’    TO #RT  /* 180719.2

          VALUE   5                                      /* 188167
            MOVE ’run number already used’       TO #RT  /* 188167
          VALUE   6                                      /* 192559
            MOVE ’run number not in range’       TO #RT  /* 192559
          VALUE 101 MOVE ’invalid function code’     TO #RT
          VALUE 102 MOVE ’parameter(s) missing’      TO #RT
          VALUE 111 MOVE ’Reserved NOP prefix used’  TO #RT /* E41116.3
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT
        END-DECIDE
        IF C-START-TIME = 0                         /* V 187483
          RESET #START-TIME
        ELSE                                        /* A 187483
          MOVE EDITED C-START-TIME (EM=YYYYMMDDHHIISS) TO #START-TIME
        END-IF                                      /* 187483
      END-REPEAT                         /* A (E-0001)
*
    VALUE ’CO’       /* -- CONDITIONS --
      RESET C-RC #RT
      R-COND. REPEAT
        INPUT ’Condition Handling’(I) #PRG-COND (AD=OD IP=OFF)
          // ’Function         ==> ’ C-FUNC (AD=MIT’_’)
          ’(S set, R reset, T test, ’’ ’’ end)’
          /  ’Owner            ==> ’ C-OWNER         (AD=MIT’_’)
          /  ’Network          ==> ’ C-NETWORK       (AD=MIT’_’)
          /  ’Condition        ==> ’ C-CONDITION     (AD=MIT’_’)
          /  ’Run Number from  ==> ’ C-RUN
          (NL=5 ZP=OFF AD=MILT’_’)
          /  ’Run Number to    ==> ’ C-RUN-TO     /* 206882.1
          (NL=5 ZP=OFF AD=MILT’_’)
          /  ’F.Date YYYYMMDD  ==> ’         /* 174441
          #DATE-FROM (AD=MIT’_’ EM=YYYYMMDD) /* 174441
          /  ’F.Time HH:II:SS  ==> ’ C-COND-TIME-FROM
          (AD=MIT’_’ EM=HH’:’II’:’SS)
          /  ’T.Date YYYYMMDD  ==> ’         /* 174441
          #DATE-TO   (AD=MIT’_’ EM=YYYYMMDD) /* 174441
          /  ’T.Time HH:II:SS  ==> ’ C-COND-TIME-TO
          (AD=MIT’_’ EM=HH’:’II’:’SS)
          /  ’conf. deletions  ==> ’ C-CONFIRM  (AD=MIT’_’)
          ’(Y/N)’
          /  ’JCL check only   ==> ’ C-JCL-CHECK  (AD=MIT’_’)
          ’(’’ ’’/C)’
          // ’Return Code    ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’ ESCAPE BOTTOM (R-COND.)  END-IF
        ADD #DATE-FROM TO C-COND-TIME-FROM     /* (E-0003)
        ADD #DATE-TO   TO C-COND-TIME-TO       /* (E-0003)
        CALLNAT #PRG-COND
          C-FUNC C-RC C-OWNER
          C-NETWORK C-CONDITION
          C-RUN   C-RUN-TO (AD=O)                /* 206882.1
          C-COND-TIME-FROM
          C-COND-TIME-TO
          C-CONFIRM (AD=O) C-JCL-CHECK (AD=O)    /* 206882.1
        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE   0 MOVE ’ok’                      TO #RT
          VALUE   1 MOVE ’condition not found’     TO #RT
          VALUE   2 MOVE ’date or run required’    TO #RT
          VALUE 101 MOVE ’invalid function code’   TO #RT
          VALUE 102 MOVE ’condition name missing’  TO #RT
          VALUE 103 MOVE ’invalid date’            TO #RT

          VALUE 104 MOVE ’network missing’         TO #RT
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT
        END-DECIDE
      END-REPEAT
*
    VALUE ’CS’       /* -- Calendars / schedules --
      RESET C-RC #RT
      R-CA. REPEAT
        INPUT ’Calendars/Schedules’(I) #PRG-CS (AD=OD IP=OFF)
          // ’Function         ==> ’ C-FUNC (AD=MIT’_’)
          ’(S set, R reset, T test, N next, P previous’
          / 26T ’1 set all, 2 reset all’
          / 26T ’C used calendar, D used schedule, ’’ ’’ end)’
          /  ’Object Type      ==> ’ C-OBJECT-TYPE   (AD=MIT’_’)
          ’(C calendar, S schedule, H schedule history)’
          /  ’Owner            ==> ’ C-OWNER         (AD=MIT’_’)
          /  ’Object           ==> ’ C-OBJECT        (AD=MIT’_’)
          /  ’Date             ==> ’ C-DATE-A8       (AD=MIT’_’)
          ’(YYYYMMDD)’
          // ’Return Code    ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’ ESCAPE BOTTOM (R-CA.)  END-IF
        CALLNAT #PRG-CS
          C-FUNC C-RC C-OBJECT-TYPE C-OWNER
          C-OBJECT C-DATE-A8
        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE   0 MOVE ’ok; date is set’                TO #RT
          VALUE   1 MOVE ’date is not set’                TO #RT
          VALUE   2 MOVE ’object not found’               TO #RT
          VALUE 101 MOVE ’invalid function code’          TO #RT
          VALUE 102 MOVE ’parameter(s) missing’           TO #RT
          VALUE 103 MOVE ’invalid parameter combination’  TO #RT
          VALUE 104 MOVE ’invalid object type’            TO #RT
          VALUE 105 MOVE ’invalid date’                   TO #RT
          VALUE 106 MOVE ’table of explicit dates full’   TO #RT
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT
        END-DECIDE
      END-REPEAT
*
    VALUE ’JI’       /* -- JOB IMPORT --
      RESET C-RC #RT
      R-JI. REPEAT
        INPUT ’Job Import’(I) #PRG-JI   (AD=OD IP=OFF)
          // ’Function       ==> ’ C-FUNC (AD=MIT’_’)
          25T ’(I import without prerequisite check’       /* 162947
          / 26T ’J import with prerequisite check’     /* 162947
          / 26T ’’’ ’’ end)’                           /* 162947
          /  ’Owner          ==> ’ C-OWNER          (AD=MIT’_’)
          /  ’Network        ==> ’ C-NETWORK        (AD=MIT’_’)
          /  ’Job            ==> ’ C-JOB            (AD=MIT’_’)
          /  ’Run            ==> ’ C-RUN            (AD=MILT’_’)
          /  ’Execution Node ==> ’ C-JI-EXEC-NODE   (AD=MILT’_’)
          /  ’Job Id         ==> ’ C-JOB-ID-10      (AD=MIT’_’)
          /  ’OS Job Name    ==> ’ C-OS-JOB-NAME    (AD=MIT’_’)

* SJU 01-06-15 (210679) START INSERTING

          /  ’Sysout File    ==> ’ C-SYSOUT-FILE-54 (AD=MIT’_’)
          /  ’Sysout Node    ==> ’ C-SYSOUT-NODE  (AD=MILT’_’)

          /  ’BS2000 Job Var ==> ’ C-BS2000-MONJV-54 (AD=MIT’_’)

* SJU 01-06-15 (210679) END   INSERTING

          // ’Return Code    ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’ ESCAPE BOTTOM (R-JI.)  END-IF
        CALLNAT #PRG-JI
          C-FUNC C-RC C-OWNER C-NETWORK C-JOB C-RUN
*         C-EXECUTION-NODE C-JOB-ID C-OS-JOB-NAME

* SJU 01-06-15 (210679) ABOVE LINE OUT, NEXT 2 IN

          C-JI-EXEC-NODE C-SYSOUT-NODE C-SYSOUT-FILE
          C-BS2000-MONJV C-JOB-ID-10 C-OS-JOB-NAME

        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE   0 MOVE ’ok’                        TO #RT
          VALUE   1 MOVE ’owner, net, job not found’ TO #RT
          VALUE 101 MOVE ’invalid function code’     TO #RT
          VALUE 102 MOVE ’parameter(s) missing’      TO #RT
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT
        END-DECIDE
      END-REPEAT
*
    VALUE ’RM’       /* -- Resource Master --
      RESET C-RC #RT
      R-RSC. REPEAT
        INPUT ’Resource Master’(I) #PRG-RSC  (AD=OD IP=OFF)
          // ’Function       ==> ’ C-FUNC (AD=MIT’_’)
          ’(A add, D delete, M modify, T test, ’’ ’’ end)’
          /  ’Node           ==> ’ C-EXECUTION-NODE (AD=MILT’_’)
          /  ’Resource       ==> ’ C-RESOURCE       (AD=MIT’_’)
          /  ’Type           ==> ’ C-RESOURCE-TYPE  (AD=MIT’_’)
          /  ’Init. Qty.     ==> ’ C-INIT-QTY-P     (AD=MILT’_’)
          /  ’Quantity       ==> ’ C-QTY-P          (AD=MILT’_’)
          // ’Return Code    ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’ ESCAPE BOTTOM (R-RSC.)  END-IF
        CALLNAT #PRG-RSC
          C-FUNC C-RC C-EXECUTION-NODE C-RESOURCE C-RESOURCE-TYPE
          C-INIT-QTY-P C-QTY-P
        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE   0 MOVE ’ok’                        TO #RT
          VALUE   1 MOVE ’resource not found’        TO #RT
          VALUE   2 MOVE ’resource existing’         TO #RT
          VALUE   3 MOVE ’inconsistent values’       TO #RT
          VALUE 101 MOVE ’invalid function code’     TO #RT
          VALUE 102 MOVE ’parameter(s) missing’      TO #RT
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT       /* (118671)
        END-DECIDE
      END-REPEAT         /* R-RSC.
*
    VALUE ’RU’       /* -- Resource Usage --           /* V E41102
      RESET C-RC #RT
      C-JOB-EXECUTED := TRUE
      R-RU. REPEAT
        INPUT ’Resource Usage’(I) #PRG-RU   (AD=OD IP=OFF)

          // ’Function       ==> ’ C-FUNC (AD=MIT’_’)
          / 20X ’(A add,’
          / 20X ’ R rel.normal, N rel.network, F rel.forced,’
          / 20X ’ ’’ ’’ end)’
          /  ’DbEnv          ==> ’ C-DBENV          (AD=MIT’_’)
          /  ’Owner          ==> ’ C-OWNER          (AD=MIT’_’)
          /  ’Network        ==> ’ C-NETWORK        (AD=MIT’_’)
          /  ’Run            ==> ’ C-RUN-4     (NL=5 AD=MILT’_’)
          /  ’Job            ==> ’ C-JOB            (AD=MIT’_’)
          /  ’Resource       ==> ’ C-RESOURCE       (AD=MIT’_’)
          /  ’Quantity       ==> ’ C-QTY-N          (AD=MILT’_’)
          /  ’Deallocation   ==> ’ C-DEALLOCATION   (AD=MIT’_’)
          ’(J after job term, N after nw term, K keep)’
          /  ’Job executed   ==> ’ C-JOB-EXECUTED
          (AD=MIT’_’ EM=N/Y)
          // ’Return Code    ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’ ESCAPE BOTTOM (R-RU.)  END-IF
        CALLNAT #PRG-RU
          C-FUNC C-RC C-DBENV C-OWNER C-NETWORK C-RUN-4 C-JOB
          C-RESOURCE C-QTY-N C-DEALLOCATION C-JOB-EXECUTED
        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE   0 MOVE ’ok’                         TO #RT
          VALUE   1 MOVE ’job active not found’       TO #RT
          VALUE   2 MOVE ’res. not defined for job’   TO #RT
          VALUE   3 MOVE ’res. definition not found’  TO #RT
          VALUE   4 MOVE ’quantity not available’     TO #RT
          VALUE 101 MOVE ’invalid function code’      TO #RT
          VALUE 102 MOVE ’parameter(s) missing’       TO #RT
          VALUE 103 MOVE ’wildcard not allowed’       TO #RT
          VALUE 104 MOVE ’deallocation wrong’         TO #RT
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT       /* (118671)
        END-DECIDE
      END-REPEAT                         /* A (109309)
*
    VALUE ’ST’       /* -- network / job status --
      RESET C-RC #RT
      R-STAT. REPEAT
        INPUT ’Network / Job Status’(I) #PRG-STAT (AD=OD IP=OFF)
          // ’Function       ==> ’ C-FUNC (AD=MIT’_’)
          ’(S status, N status of next, R get latest run’
          / 24T ’A get next run, Y get symbol table, ’’ ’’ end)’
          /  ’Owner          ==> ’ C-OWNER         (AD=MIT’_’)
          /  ’Network        ==> ’ C-NETWORK       (AD=MIT’_’)
          /  ’Run            ==> ’ C-RUN
          (NL=5 ZP=OFF AD=MILT’_’)
          /  ’Job            ==> ’ C-JOB           (AD=MIT’_’)
          /  ’Symbol Table   ==> ’ C-SYMBOL-TABLE  (AD=OD)
          // ’Return Code    ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’  ESCAPE BOTTOM (R-STAT.)  END-IF
        CALLNAT #PRG-STAT
          C-FUNC C-RC C-OWNER C-NETWORK C-JOB C-RUN
          C-SYMBOL-TABLE
        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE 0   MOVE ’function performed’        TO #RT
          VALUE 1   MOVE ’network/job not found’     TO #RT
          VALUE 20  MOVE ’to be activated’           TO #RT
          VALUE 21  MOVE ’waiting for activation’    TO #RT

          VALUE 25  MOVE ’waiting for prerequisite’  TO #RT
          VALUE 26  MOVE ’one or more jobs in hold’  TO #RT
          VALUE 41  MOVE ’executing’                 TO #RT
          VALUE 65  MOVE ’all jobs ended ok’         TO #RT
          VALUE 66  MOVE ’any job ended not ok’      TO #RT
          VALUE 69  MOVE ’any permanent error’       TO #RT
          VALUE 93  MOVE ’awaiting deactivation’     TO #RT
          VALUE 101 MOVE ’invalid function code’     TO #RT
          VALUE 102 MOVE ’parameter(s) missing’      TO #RT
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT       /* (118671)
        END-DECIDE
        END TRANSACTION           /* (103331)
      END-REPEAT
*
    VALUE ’SW’       /* -- next Job wait Status /Error --  V 207190
      RESET C-RC #RT
      R-STSW. REPEAT
        INPUT ’Next Job wait status / error’(I) #PRG-STE (AD=OD IP=OFF)
          // ’Function       ==> ’ C-FUNC (AD=MIT’_’)
          ’(S Status, N Status of next, ’
          / 24T ’W Next Job wait status / error, R get latest run’
          / 24T ’A get next run, Y get symbol table, ’’ ’’ end)’
          /  ’Owner          ==> ’ C-OWNER         (AD=MIT’_’)
          /  ’Network        ==> ’ C-NETWORK       (AD=MIT’_’)
          /  ’Run            ==> ’ C-RUN
          (NL=5 ZP=OFF AD=MILT’_’)
          /  ’Job            ==> ’ C-JOB           (AD=MIT’_’)
          /  ’Symbol Table   ==> ’ C-SYMBOL-TABLE  (AD=OD)
          /  ’Job ID         ==> ’ C-JOB-ID-10     (AD=OD)
          /  ’Status Time    ==> ’ C-STATUS-TIME   (AD=OD)
          // ’Return Code    ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’  ESCAPE BOTTOM (R-STSW.)  END-IF
        CALLNAT #PRG-STE
          C-FUNC C-RC C-OWNER C-NETWORK C-JOB C-RUN
          C-SYMBOL-TABLE C-JOB-ID-10 C-STATUS-TIME
        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE 0   MOVE ’function performed’              TO #RT
          VALUE 1   MOVE ’network/job not found’           TO #RT
          VALUE 20  MOVE ’to be activated’                 TO #RT
          VALUE 21  MOVE ’waiting for activation’          TO #RT
          VALUE 25  MOVE ’waiting for prerequisite’        TO #RT
          VALUE 26  MOVE ’one or more jobs in hold’        TO #RT
          VALUE 27  MOVE ’Wait for next action/start time’ TO #RT
          VALUE 28  MOVE ’Wait for Condition’              TO #RT
          VALUE 29  MOVE ’Wait for Resource’               TO #RT
          VALUE 30  MOVE ’Wait for Opsys Server’           TO #RT
          VALUE 31  MOVE ’JCL to be loaded’                TO #RT
          VALUE 32  MOVE ’NW activation error’             TO #RT
          VALUE 33  MOVE ’Job activation error’            TO #RT
          VALUE 34  MOVE ’Symbol replacement error’        TO #RT
          VALUE 35  MOVE ’Any Extraction+Activation error’ TO #RT
          VALUE 36  MOVE ’Job submission’                  TO #RT
          VALUE 37  MOVE ’Job submission error’            TO #RT
          VALUE 38  MOVE ’Job execution’                   TO #RT
          VALUE 39  MOVE ’Job execution error’             TO #RT
          VALUE 41  MOVE ’executing’                       TO #RT
          VALUE 42  MOVE ’End of Job Checking’             TO #RT
          VALUE 43  MOVE ’End of Job Checking error’       TO #RT
          VALUE 44  MOVE ’End of Job Actions’              TO #RT

          VALUE 45  MOVE ’End of Job Actions error’        TO #RT
          VALUE 65  MOVE ’all jobs ended ok’               TO #RT
          VALUE 66  MOVE ’any job ended not ok’            TO #RT
          VALUE 69  MOVE ’any permanent error’             TO #RT
          VALUE 93  MOVE ’awaiting deactivation’           TO #RT
          VALUE 101 MOVE ’invalid function code’           TO #RT
          VALUE 102 MOVE ’parameter(s) missing’            TO #RT
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT
        END-DECIDE
        END TRANSACTION
      END-REPEAT                                   /* A 207190
*

    VALUE ’SY’       /* -- SYMBOLS --
      RESET C-RC #RT
      R-SYM. REPEAT
        INPUT ’Symbol Handling’(I) #PRG-SYM (AD=OD IP=OFF)
          // ’Function       ==> ’ C-FUNC (AD=MIT’_’)
          ’(S set, R reset, T test, N next,’
          / 30T ’M add mult, D delete mult, ’’ ’’ end)’
          /  ’Owner          ==> ’ C-OWNER         (AD=MIT’_’)
          /  ’Network        ==> ’ C-NETWORK       (AD=MIT’_’)
          ’(active only)’
          /  ’Run            ==> ’ C-RUN
          (NL=5 ZP=OFF AD=MILT’_’)
          ’(active only)’
          /  ’Symbol Table   ==> ’ C-SYMBOL-TABLE  (AD=MIT’_’)
          /  ’Symbol Name    ==> ’ C-SYMBOL-NAME   (AD=MIT’_’)
          /  ’Symbol Format  ==> ’ C-SYMBOL-FORMAT (AD=MIT’_’)
          /  ’Symbol Value   ==> ’ C-SYMBOL-VALUE-1  (AD=MI’_’)
          /  ’                   ’ C-SYMBOL-VALUE-2  (AD=MI’_’)
          // ’modified       ==> ’ C-MOD-USER (AD=OD)
          C-MOD-TIME (AD=OD IP=OFF EM=YY’-’MM’-’DD’ ’HH’:’II’:’SS)
          // ’Return Code    ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’  ESCAPE BOTTOM (R-SYM.)  END-IF
        CALLNAT #PRG-SYM
          C-FUNC C-RC C-OWNER C-NETWORK C-RUN
          C-SYMBOL-TABLE C-SYMBOL-NAME C-SYMBOL-FORMAT C-SYMBOL-VALUE
          C-MOD-USER C-MOD-TIME
        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE 0   MOVE ’ok’                        TO #RT
          VALUE 1   MOVE ’symbol not found’          TO #RT
          VALUE 2   MOVE ’invalid numeric value’     TO #RT
          VALUE 3   MOVE ’invalid format’            TO #RT
          VALUE 4   MOVE ’format change attempted’   TO #RT
          VALUE 5   MOVE ’mult. table full’          TO #RT /* (103331)
          VALUE 6   MOVE ’value missing’             TO #RT /* (103331)
          VALUE 7   MOVE ’value not found’           TO #RT /* (103331)
          VALUE 10
            #RT := ’rejected by userexit’
          VALUE 11  MOVE ’userexit not found’        TO #RT /* (107759)
          VALUE 12  MOVE ’access denied by userexit’ TO #RT /* (E-0029)
          VALUE 20  MOVE ’ok, is multiple value’     TO #RT /* (921229)
          VALUE 30
            #RT := ’ok, master updated too’          /* 158341
          VALUE 101 MOVE ’invalid function code’     TO #RT
          VALUE 102 MOVE ’parameter(s) missing’      TO #RT /* 960301
          VALUE 111 MOVE ’Reserved NOP prefix used’  TO #RT /* E41116.3
          NONE VALUE

            COMPRESS ’response’ C-RC INTO #RT       /* (118671)
        END-DECIDE
      END-REPEAT
*
    VALUE ’VE’       /* -- VERSION INFO --
      MOVE ’V’ TO C-FUNC
      RESET C-VERSION C-VERSION-DATE
      CALLNAT #PRG-VERS
        C-FUNC C-VERSION C-VERSION-DATE C-UPDATE-DATE
      WRITE ’Entire Operations’ C-VERSION
        / ’Version Date     ’ C-VERSION-DATE
        / ’Update  Date     ’ C-UPDATE-DATE
*
    VALUE ’SP’       /* -- Symbol Prompting 1246247
      RESET C-RC #RT C-PROMPT-TEXT (*)
      R-SYP. REPEAT
        INPUT ’Symbol Prompting’(I) #PRG-PT (AD=OD IP=OFF)
          // ’Owner          ==> ’ C-OWNER         (AD=MIT’_’)
          /  ’Symbol Table   ==> ’ C-SYMBOL-TABLE  (AD=MIT’_’)
          /  ’Symbol Name    ==> ’ C-SYMBOL-NAME   (AD=MIT’_’)
          // ’Return Code    ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
          /  ’Prompt Text 1  ==> ’ C-PROMPT-TEXT (1) (AD=OD)
          /  ’Prompt Text 2  ==> ’ C-PROMPT-TEXT (2) (AD=OD)
          /  ’Prompt Text 3  ==> ’ C-PROMPT-TEXT (3) (AD=OD)
          /  ’Prompt Text 4  ==> ’ C-PROMPT-TEXT (4) (AD=OD)
          /  ’Prompt Text 5  ==> ’ C-PROMPT-TEXT (5) (AD=OD)
        IF C-OWNER = ’ ’ OR C-SYMBOL-TABLE = ’ ’
            OR C-SYMBOL-NAME= ’ ’
          ESCAPE BOTTOM (R-SYP.)
        END-IF
        CALLNAT #PRG-PT
          C-OWNER C-SYMBOL-TABLE C-SYMBOL-NAME
          C-RC C-PROMPT-TEXT (*)
        END TRANSACTION
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE 0
            MOVE ’ok’                        TO #RT
          VALUE 1
            MOVE ’symbol not found’          TO #RT
          VALUE 111                                 /* E41116.3
            MOVE ’Reserved NOP prefix used’  TO #RT /* E41116.3
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT       /* (118671)
        END-DECIDE
      END-REPEAT
*
    VALUE ’SC’                                      /* V 183226 add
      RESET C-RC #RT
      R-SC. REPEAT
        INPUT ’Job schedule’(I) #PRG-SC (AD=OD IP=OFF)
          // ’Function        ==> ’ C-FUNC (AD=MIT’_’)
          ’(R inquire, N inquire next, S change job schedule)’
          /  ’Owner           ==> ’ C-OWNER     (AD=MIT’_’)
          /  ’Network         ==> ’ C-NETWORK   (AD=MIT’_’)
          /  ’Run Number      ==> ’ C-RUN
          (NL=5 ZP=OFF AD=MILT’_’)
          /  ’Job             ==> ’ C-JOB       (AD=MIT’_’)
          //  ’Earliest-start  ==> ’ C-EARLIEST-START (AD=MIT’_’)
          ’In/Out (YYYYMMDDHHIISS) Active’
          /  44T ’(00000000HHIISS) Master’
          //  ’Latest-start    ==> ’ C-LATEST-START   (AD=MIT’_’)

          ’In/Out (YYYYMMDDHHIISS) Active’
          /  44T ’(00000NNNHHIISS) Master’
          //  ’Deadline        ==> ’ C-DEADLINE       (AD=MIT’_’)
          ’In/Out (YYYYMMDDHHIISS) Active’
          /  44T ’(00000NNNHHIISS) Master’
          //  ’Average-time    ==> ’ C-AVERAGE-TIME   (ZP=OFF AD=MIT’_’)
          ’   In(estimated)/Out(average) 1/10 of Second’
          // ’Return-code     ==> ’ C-RC (AD=OD) ’ ’ #RT (AD=OD)
        IF C-FUNC = ’ ’ OR = ’.’
          ESCAPE BOTTOM (R-SC.)
        END-IF
        CALLNAT #PRG-SC C-FUNC (AD=O) C-RC C-OWNER (AD=O)
          C-NETWORK (AD=O) C-RUN (AD=O) C-JOB
          C-AVERAGE-TIME C-EARLIEST-START
          C-LATEST-START C-DEADLINE
*
        DECIDE ON FIRST VALUE OF C-RC
          VALUE   0 MOVE ’ok’                                 TO #RT
          VALUE   1 MOVE ’owner,network,run or job not found’ TO #RT
          VALUE 101 MOVE ’invalid function code             ’ TO #RT
          VALUE 102 MOVE ’parameter missing                 ’ TO #RT
          VALUE 103 MOVE ’Wrong Earliest Start              ’ TO #RT
          VALUE 104 MOVE ’Wrong Latest Start                ’ TO #RT
          VALUE 105 MOVE ’Wrong Deadline                    ’ TO #RT
          VALUE 106 MOVE ’Latest must be> Earliest Start   ’ TO #RT
          VALUE 107 MOVE ’Deadline must be> Latest Start   ’ TO #RT
          VALUE 108 MOVE ’Job already Submitted             ’ TO #RT
          VALUE 109 MOVE ’Earliest must be> Current Time   ’ TO #RT
*                                               183994 added
          VALUE 110 MOVE ’Job in Hold                       ’ TO #RT
          NONE VALUE
            COMPRESS ’response’ C-RC INTO #RT
        END-DECIDE
      END-REPEAT                                    /* A 183226
*
    VALUE ’LM’     /* log monitor activity (Y/N)        /* V 190708.2
      R-LM. REPEAT
        INPUT ’Log monitor activity’(I) #PRG-MLA (AD=OD IP=OFF)
          // ’Function        ==> ’ C-FUNC (AD=MIT’_’)
          ’(Y yes, N no)’
        IF C-FUNC = ’ ’ OR = ’.’
          ESCAPE BOTTOM (R-LM.)
        END-IF
        CALLNAT #PRG-MLA C-FUNC (AD=O)
      END-REPEAT                                    /* A 190708.2
*
    VALUE ’AS’                 /* act job stats sju 01-06-08 (080601) V
      P-FUNCTION:= ’S’
      CALLNAT #PRG-AJS NOPUAS-VERSION NOPUAS-PARMS(*)    /* 011118 SN
      WRITE ’Jobs waiting.......:’ P-NUMBER-WAITING
        / ’Jobs in input queue:’ P-NUMBER-INPUT-QUEUE
        / ’Jobs held..........:’ P-NUMBER-HOLD
        / ’Jobs executing.....:’ P-NUMBER-EXECUTING
        / ’Jobs in error......:’ P-NUMBER-PERM-ERROR    /* (080601) A
    NONE VALUE
      WRITE ’Invalid Selection’
  END-DECIDE
END-REPEAT
END

